/**
 * SPDX-License-Identifier: MIT
 */ 

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}


pragma solidity ^0.8.0;


// CAUTION
// This version of SafeMath should only be used with Solidity 0.8 or later,
// because it relies on the compiler's built in overflow checks.

/**
 * @dev Wrappers over Solidity's arithmetic operations.
 *
 * NOTE: `SafeMath` is generally not needed starting with Solidity 0.8, since the compiler
 * now has built in overflow checking.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            uint256 c = a + b;
            if (c < a) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the substraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b > a) return (false, 0);
            return (true, a - b);
        }
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
            // benefit is lost if 'b' is also tested.
            // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
            if (a == 0) return (true, 0);
            uint256 c = a * b;
            if (c / a != b) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a / b);
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a % b);
        }
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        return a + b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        return a * b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator.
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b <= a, errorMessage);
            return a - b;
        }
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a / b;
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a % b;
        }
    }
}

// File: wagmi staking/Context.sol


pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}
// File: wagmi staking/Ownable.sol


pragma solidity ^0.8.0;


/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}
// File: wagmi staking/wagmi_staking.sol


pragma solidity ^0.8.17;



abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

/**
 * BEP20 standard interface.
 */
interface IERC20 {
    function totalSupply() external view returns (uint256);
    function decimals() external view returns (uint8);
    function symbol() external view returns (string memory);
    function name() external view returns (string memory);
    function getOwner() external view returns (address);
    function balanceOf(address account) external view returns (uint256);
    function transfer(address recipient, uint256 amount) external returns (bool);
    function allowance(address _owner, address spender) external view returns (uint256);
    function approve(address spender, uint256 amount) external returns (bool);
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);
    event Transfer(address indexed from, address indexed to, uint256 value);
    event Approval(address indexed owner, address indexed spender, uint256 value);
}


contract WagmigamesLock is Ownable, ReentrancyGuard {
    using SafeMath for uint256;

	struct TPlan {
		uint256 durationDays;
		uint256 investFactor;
	}

	struct TDeposit {
		uint256 planIdx;
		uint256 amount;
		uint256 timeStart;
		uint256 timeEnd;
		uint256 profit;
		uint256 checkpoint;
		uint256 depositIdx;
        uint256 claimedRewards;
        uint256 compoundedRewards;
		bool isDeceased;
	}

	struct TUser {
		TDeposit[] deposits;
		uint256 totalInvested;
		uint256 totalClaimed;
        uint256 totalCompounded;
	}

    mapping(address => bool) public _isBlacklisted;
	mapping( address => TUser ) public users;
	TPlan[] public plans;

    uint256 public constant TIME_STEP = 1 days;
    // uint256 public constant TIME_STEP = 1 minutes; // test time step
    uint256 public constant PERCENTS_DIVIDER = 1000;
   
    uint256 public totalDepositNo;
	uint256 public totalInvested;
	uint256 public totalClaimed;
    uint256 public totalCompounded;

    bool public launched;
    bool public emergencyPaused = false;

    address public stakingAddress = 0x3B604747ad1720C01ded0455728b62c0d2F100F0;

    IERC20 stakingToken = IERC20(stakingAddress);

	event Claimed(address indexed user, uint256 amount,  uint256 claimedTime);
    event Compounded(address indexed user, uint256 amount, uint256 compoundedTime);
    event Withdrawn(address indexed user,  uint256 amount, uint256 withdrawnTime);
	event NewDeposit(address indexed user, uint256 planIdx, uint256 amount);
    event ERC20TokensRemoved(address indexed tokenAddress, address indexed receiver, uint256 amount);

    constructor(){
        // // unlocked plans
        plans.push( TPlan(75, 35 ) ); // 3.5%
        // locked plans
		plans.push( TPlan(30, 40 ) ); // 4%
        plans.push( TPlan(90, 65 ) ); // 6.5
        plans.push( TPlan(365, 100 ) ); // 10%
    }


    /**
     * add stake token to staking pool
     * @dev requires the token to be approved for transfer
     * @dev we assume that (our) stake token is not malicious, so no special checks
     * @param _amount of token to be staked
     */

    function lockup(uint256 _amount, uint8 _planIdx) external nonReentrant {
        require(!emergencyPaused, "Project has been paused");
        require (launched, "Project is not launched.");
        require(_amount > 0, "Deposit amount can't be zero");
        require(_planIdx < plans.length, "Invalid plan index");
        require(!_isBlacklisted[msg.sender], "Blacklisted address");

        // Transfer the tokens for staking
        stakingToken.transferFrom(msg.sender, address(this), _amount);
        
        _createDeposit( msg.sender, _planIdx, _amount);

	}

    function _createDeposit(address _user, uint256 _planIdx, uint256 _amount)
		internal
		returns(uint256 o_depIdx)
	{

		TUser storage user = users[_user];
		TDeposit memory newDep;

		(uint256 profit) = _getResult(_planIdx, _amount);

		o_depIdx = user.deposits.length;
		newDep = TDeposit(
			_planIdx,
			_amount,
			block.timestamp,
			block.timestamp + plans[_planIdx].durationDays * TIME_STEP,
			profit,
			block.timestamp,
			o_depIdx,
            0,
            0,
			false
			);

		user.deposits.push(newDep);

		user.totalInvested += _amount;
		totalDepositNo++;
		totalInvested += _amount;

		emit NewDeposit(_user, newDep.planIdx, newDep.amount);
	}


    /**
     * Withdraw staked funds from the contract
     * @dev must be called only when `block.timestamp` >= `lockupPeriod`
     * @dev `block.timestamp` higher than or equal to `lockupPeriod` (lockupPeriod finished)
     */

    
    function withdraw(uint256 depIdx) public nonReentrant{
        require(!emergencyPaused, "Project has been paused");
        require(!_isBlacklisted[msg.sender], "Blacklisted address");
        unlock(depIdx);
    }

    /**
    * Withdraw just locked amounts in case of project emergency
    */

    function emergencyWithdraw(uint256 depIdx) public nonReentrant {
        require(emergencyPaused, "Project has been paused");

        TUser storage user = users[msg.sender];

        require(!user.deposits[depIdx].isDeceased, "deposit is deceased");

        user.deposits[depIdx].checkpoint = block.timestamp;
		user.deposits[depIdx].isDeceased = true;

         uint256 withdrawableAmount = user.deposits[depIdx].amount;

        uint256 balance = getContractBalance();
		if (withdrawableAmount > balance) {
			withdrawableAmount = balance;
		}
        require(withdrawableAmount > 0, "Nothing to withdraw");

        stakingToken.transfer(msg.sender, withdrawableAmount);

        emit Withdrawn(msg.sender, withdrawableAmount,  block.timestamp);
    }

    function unlock (uint256 depIdx) private {
        TUser storage user = users[msg.sender];

		uint256 planIdx = user.deposits[depIdx].planIdx;
        uint256 lockupPeriod = user.deposits[depIdx].timeEnd;
        uint256 depositAmount = user.deposits[depIdx].amount;
        
        // Locked Packages
        if(planIdx >= 1){
            require(block.timestamp >= lockupPeriod,  "No withdraw until lockup ends");
        }
        
        require(!user.deposits[depIdx].isDeceased, "deposit is deceased");

		(uint256 claimAmount_, uint256 checkpoint_) = _calculateDepositDividends(msg.sender, depIdx);
      
        user.deposits[depIdx].checkpoint = checkpoint_;
		user.deposits[depIdx].isDeceased = true;
        user.deposits[depIdx].claimedRewards += claimAmount_;

        uint256 withdrawableAmount = depositAmount.add(claimAmount_);

        uint256 balance = getContractBalance();
		if (withdrawableAmount > balance) {
			withdrawableAmount = balance;
		}
        require(withdrawableAmount > 0, "Nothing to withdraw");

        
        user.totalClaimed += withdrawableAmount;
		totalClaimed += withdrawableAmount;

        // Send user amount staked tokens + rewards
        stakingToken.transfer(msg.sender, withdrawableAmount);
        

		emit Claimed(msg.sender, claimAmount_, block.timestamp );
        emit Withdrawn(msg.sender, withdrawableAmount,  block.timestamp);
       
    }


     /**
     * claim reward tokens for accumulated reward credits
     * ... but do not unstake staked token
     */

    function claim(uint256 depIdx) public nonReentrant{
        require(!emergencyPaused, "Project has been paused");
        TUser storage user = users[msg.sender];
        uint256 lockupPeriod = user.deposits[depIdx].timeEnd;
        uint256 claimAmount;

        // if it is passed the lockup period, then unstake deposit and rewards
        if(block.timestamp >= lockupPeriod){
                unlock(depIdx);
        }
        else{
            (uint256 claimAmount_, uint256 checkpoint_) = _calculateDepositDividends(msg.sender, depIdx);
            require(claimAmount_ > 0, "Claim Amount must be greater than zero");
            require(checkpoint_ > 0, "Checkpoint time must be greater than zero");
            
            updateCheckPoint(msg.sender, depIdx, checkpoint_);
            claimAmount =  claimAmount_;
            uint256 balance = getContractBalance();

            if (claimAmount > balance) {
                claimAmount = balance;
            }

            user.deposits[depIdx].claimedRewards += claimAmount;
            user.totalClaimed += claimAmount;
            totalClaimed += claimAmount;

            stakingToken.transfer(msg.sender, claimAmount);

            emit Claimed(msg.sender, claimAmount,block.timestamp );
        }
        
    }

    function compoundDividends(uint256 depIdx) external nonReentrant{
        require(!emergencyPaused, "Project has been paused");
        TUser storage user = users[msg.sender];
        uint256 lockupPeriod = user.deposits[depIdx].timeEnd;
        uint256 amountLocked = user.deposits[depIdx].amount;

        require(block.timestamp < lockupPeriod, "Passed the lockup period");

        (uint256 claimAmount_, uint256 checkpoint_) = _calculateDepositDividends(msg.sender, depIdx);
        require(claimAmount_ > 0, "Claim Amount must be greater than zero");
        require(checkpoint_ > 0, "Checkpoint time must be greater than zero");

        updateCheckPoint(msg.sender, depIdx, checkpoint_);
      
        // compound the claimed amount
        user.deposits[depIdx].amount = amountLocked + claimAmount_;

        user.deposits[depIdx].compoundedRewards += claimAmount_; // update the compounded rewards

        user.totalCompounded += claimAmount_;
        totalCompounded += claimAmount_;

        emit Compounded(msg.sender, claimAmount_, block.timestamp);
    }

    function updateCheckPoint(address addr, uint256 deptId, uint256 checkPoint)
		internal
	{
		TUser storage user = users[addr];

		user.deposits[deptId].checkpoint = checkPoint;

		if(checkPoint >= user.deposits[deptId].timeEnd)
			user.deposits[deptId].isDeceased = true;
	}

    function _isDepositDeceased(TUser memory user_, uint256 depositIndex) internal pure returns(bool) {
		TDeposit memory userDeposits = user_.deposits[depositIndex];

		return (userDeposits.checkpoint >= userDeposits.timeEnd);
	}

    function _getResult(
		uint256 planIdx,
		uint256 amount
	)
		private
		view
		returns
		(
			uint256 profit
		)
	{
		TPlan memory plan = plans[planIdx];

		uint256 factor = plan.investFactor;

		profit = amount.div(PERCENTS_DIVIDER).mul(factor);
	}

    function _calculateDepositDividends(address _user, uint256 _depIdx) 
    public view returns (uint256 o_amount, uint256 checkPoint) {
		TUser storage user = users[_user];

		TDeposit storage deposit = user.deposits[_depIdx];

		//calculate withdrawable dividends starting from the last Claim checkpoint
		uint256 timeA = deposit.timeStart > deposit.checkpoint ? deposit.timeStart : deposit.checkpoint;
		uint256 timeB = block.timestamp >= deposit.timeEnd ? deposit.timeEnd : block.timestamp;

		 // Let's calculate the maximum amount which can be earned per annum
        (uint256 interestEarnedPerAnnum) = _getResult(deposit.planIdx, deposit.amount);
        uint256 dayCounts = plans[deposit.planIdx].durationDays * TIME_STEP;

        // Let's calculate how many wei are earned per second
        uint256 weiPerSecond = interestEarnedPerAnnum.div(dayCounts);
        
        if (timeA < timeB) {
            uint256 secondsRemaining = timeB.sub(timeA);
            // There are 31536000 seconds per annum, so let's calculate the interest for this remaining time period
            o_amount = weiPerSecond.mul(secondsRemaining);
            checkPoint = block.timestamp;
        }
	}

    
    function blacklistAddress(address account, bool value) external onlyOwner{
        _isBlacklisted[account] = value;
    }

    function launch() external onlyOwner() {
		launched = true;
	}

    function setEmergencyPause(bool value) external onlyOwner(){
        emergencyPaused = value;
    }

     /**
     * @notice withdraw accidently sent ERC20 tokens
     * @param _tokenAddress address of token to withdraw
     */
    function removeOtherERC20Tokens(address _tokenAddress) external onlyOwner() {
        require(_tokenAddress != address(stakingToken), "can not withdraw staking token");
        uint256 balance = IERC20(_tokenAddress).balanceOf(address(this));
        IERC20(_tokenAddress).transfer(msg.sender, balance);
        
        emit ERC20TokensRemoved(_tokenAddress, msg.sender, balance);
    }


    function getPackageInfo(uint256 index_) external view returns(TPlan memory) {
		return plans[index_];
	}

	function getProjectInfo()
		external
		view
		returns(
			uint256 o_totDeposits,
			uint256 o_totInvested,
			uint256 contractBalance,
			uint256 o_timestamp
		)
	{
		return( totalDepositNo, totalInvested, getContractBalance(), block.timestamp );
	}

     /**
     * @return balance of reward tokens held by this contract
     */
    function getContractBalance() public view returns (uint256 balance) {
        if (stakingAddress == address(0)) return 0;
        balance = stakingToken.balanceOf(address(this));
    }

    function getUserDeposits(address _user)
		external
		view
		returns(TDeposit[] memory)
	{
		TUser storage user = users[_user];

		return user.deposits;
	}

	function getUserInfo(address _user)
		external
		view
		returns(
			uint256 stakedAmount,
			uint256 availableAmount
		)
	{
		TUser storage user = users[_user];


		stakedAmount = user.totalInvested;

		uint256 claimAmount;

		for(uint256 i = 0; i < user.deposits.length; i++) {
			if(_isDepositDeceased(user,i)) continue;
			// if(user.deposits[i].planIdx >= 1) continue; // remove this line

			(uint256 claimAmount_, uint256 checkpoint_) = _calculateDepositDividends(_user,i);

			if(claimAmount_ <= 0) continue;
			if(checkpoint_ <= 0) continue;

			claimAmount += claimAmount_;
		}

		availableAmount = claimAmount;
	}

}